
Abstract 
By using the polarization effects induced by a 

ferroelectric layer, a normally off AlGaN/GaN 

HEMT is achieved in simulations, with minimal 

leakage current and a high current capability. 

1. Introduction 
When it comes to the AlGaN/GaN HEMT, 

polarization plays a key role in its functioning. 

This is because the polarization effects at the 

AlGaN/GaN interface causes a two-dimensional 

electron gas (2DEG) to form just below the 

interface in the gallium-nitride. GaN-based 

HEMTs are interesting for their high speed and 

high-power handling capabilities. This is due to 

the high electron mobility, which is caused by the 

2DEG, hence the name high electron mobility 

transistor (HEMT). The one downside is that a 

conventional version is normally-on due to the 

strong polarization at the AlGaN/GaN interface. 

This makes them less practical due to the need of 

a more complex driver circuit and potential 

hazards.  

There are already multiple methods of achieving 

a normally off GaN HEMT, each with their own up 

and downsides. Some methods have been 

studied more than others, these are the cascode 

configuration, p-GaN, doping under the gate and 

recessed gate (1). Another method is by using a 

ferroelectric in the gate stack. This method makes 

use of the polarization properties of a 

ferroelectric material to counteract the 

polarization already present at the AlGaN/GaN 

interface, therefore depleting the 2DEG. By 

depleting the 2DEG the high-power capacity of 

the HEMT stays relatively unchanged. This is 

opposed to destroying the 2DEG, which happens 

with for example the recessed gate. This 

destruction creates a gap in the 2DEG, which 

results in a higher channel resistance, thus 

making this less suitable for high power 

applications. 

In this paper, a feasibility study of adopting a 

ferroelectric material in the gate stack to obtain a 

normally off/enhanced mode (e-mode) 

AlGaN/GaN HEMT has been carried out. This 

study has been carried out by performing 

Figure 1: Structure of the final design. 

computer (FEM) simulations with a commercial 

two-dimensional device simulation tool, Atlas (2). 

The device optimization process will be explored 

in the following section, followed by the obtained 

results, a discussion of these results, and finally 

the conclusion with some future 

recommendations.  

2. Design optimization process 
The design process was carried out in Silvaco 

Atlas, which is a two-dimensional device 

simulator. The final design can be found in figure 

1. The gate stack can be seen in more detail in 

figure 2. The gate stack consists of two layers on 

top of the AlGaN, the PZT, which is the 

ferroelectric, and underneath that is a layer of 

hafnium dioxide. This layer of hafnium dioxide is 

chosen as a charge trapping layer (3). The rest of 

the structure is the same as a conventional 

AlGaN/GaN HEMT, an AlGaN layer on top of a 

GaN layer, with an insulating oxide layer at the 

bottom. In the simulations the source and drain 

are recessed so they reach into the GaN and 

therefore connect to the 2DEG, and the gate 

electrode rests on top of the gate stack, on the 

PZT. The thicknesses of the layers are 30𝑛𝑚 for 

the PZT and hafnium oxide, 20𝑛𝑚 for the AlGaN, 

and 1.5µ𝑚 for the GaN. The length of the whole 

HEMT is 9µ𝑚, with a channel length of 7µ𝑚. The 

gate length LG and drift length LD are 2µ𝑚 and 

3µ𝑚 respectively.

 

Figure 2: Gate stack of the final design. 

Multiple design choices were made based on 

existing research or remained unchanged from 

the template one of my supervisors provided me. 

No modifications have been made to the width of 

any of the regions or the total HEMT, and nothing 

was changed at all underneath the AlGaN layer. 

The position of the gate stack was not 

experimented with either. When it comes to the 

ferroelectric material, no material changes were 

made. The PZT that was already present in the 

template is a material that is known and proven 

to work in this role, therefore changing it was not 

a priority (4). Layer thicknesses were 

experimented with, and from the simulation 

exercise the same conclusion can be drawn as the 

literature, thinner layers work better (3) (4) (5). 

The thinner layers resulted higher threshold 

voltages, yet still negative, and steeper current 

curves at the threshold voltage. Therefore, the 

minimum thickness that Atlas would allow at this 

point before running into issues was chosen.  

Another design choice that is based on existing 

research is the use of a trapping layer between 

the ferroelectric and the AlGaN (3). Hafnium 

dioxide was picked as the material for the 

trapping layer since this was used with success in 

one of the papers (3). The decision to stick with 

this material and not experiment with other 

materials was made due to time constraints and 

because this material has been proven to work 

already. Most of the time went into finding the 

optimal polarization by varying the sheet charge 

densities. The final sheet charge values that were 

set are 8 × 1013𝑐𝑚−2 for the PZT, which is 

located at the bottom of the PZT, and −2.7 ×

1013𝑐𝑚−2 and  0.8 × 1013𝑐𝑚−2 at the top and 

bottom of the AlGaN respectively. The 

polarization in the PZT can be achieved by poling 

the material. The AlGaN values can be varied by 

adjusting the aluminium gallium composition, 

which is 0.16 to 0.84 in the final design. Table 1 

shows an overview of the device parameters. 

Although the only significant difference in the 

gate stack between the template that was 

handed over and this final design is an additional 

hafnium dioxide layer in the gate stack, more 

things were experimented with during the design 

process. After integrating the hafnium dioxide 

layer into the gate stack, the next step was 

experimenting with a recessed gate. According to 

the research a greater recess depth should result 

in a higher threshold voltage yet lower drain 



current. The impact of the recess was limited, this 

may have been due to the persistent problems 

with the meshing caused by misaligned layers. 

Although the voltage was shifted slightly, the 

amount was not significant enough to shift it to a 

positive voltage, and therefore dropped to save 

on time, avoid more problems, and avoid the 

decrease in drain current caused by it. 

Table 1: Design parameters 

 
Final design Antiparallel 

design 
conventional 

Device length (µm) 9 9 9 

Channel length (µm) 7 7 7 

Gate length (µm) 2 2 2 

Drift length (µm) 3 3 3 

PZT layer thickness 
(nm) 

30 30 - 

HfO2 layer thickness 
(nm) 

30 30 - 

AlGaN layer thickness 
(nm) 

20 20 20 

GaN layer thickness 
(µm) 

1.5 1.5 1.5 

Aluminium/Gallium 
composition 

0.16/0.84 0.16/0.84 0.16/0.84 

Sheet charge PZT/HfO2 

(𝑐𝑚−2) 
8 × 1013 −8 × 1013 - 

Sheet charge 
HfO2/AlGaN (𝑐𝑚−2) 

−2.7 × 1013 −2.7 × 1013 - 

Sheet charge 
AlGaN/GaN (𝑐𝑚−2) 

0.8 × 1013 0.8 × 1013 0.8 × 1013 

 

After reverting back to the unrecessed gate with 

the PZT/hafnium dioxide gate stack, some more 

exercises was done with it. It is here where a 

threshold voltage that approached 0 was 

achieved, which was achieved with unrealistically 

high polarization values. Getting past the 0 volts 

was the difficult part. In the following struggle to 

achieve a positive threshold voltage some things 

were tried that go against the theory. One thing 

that was tried and oddly enough yielded a 

positive threshold voltage, therefore achieving a 

normally off HEMT. This thing was changing the 

polarization of the PZT layer from antiparallel to 

parallel with the polarization at the AlGaN/GaN 

interface. This is illogical because an antiparallel 

raises the conduction band while a parallel one 

lowers it, a comparison of this can be seen in 

figure 3 and 4. The band bending is the strongest 

near the gate electrode, which is on the left, and 

the difference becomes significantly less at the 

2DEG, where the conduction bands of the parallel 

and antiparallel are 0.507𝑒𝑉 and 0.536𝑒𝑉 

respectively. Note that these figures are with the 

final values, which can be found in table 1, where 

final design is the parallel polarization and 

antiparallel design is with the antiparallel 

polarization. At this point the next task was to 

bring the polarization values back down to 

realistic values, which resulted in the final design 

described above. 

 

Figure 3: Band diagram with parallel polarization (0𝑉𝐺𝑎𝑡𝑒). 

 

Figure 4: Band diagram with antiparallel polarization 
(0𝑉𝐺𝑎𝑡𝑒). 

The structure indicate that the ferroelectric layer 

is germanium, but this is not the case. The 

material properties are the ones from PZT, the 

only reason it is named germanium is for atlas to 

treat is at a semiconductor, which is needed for it 

to work. 

3. Results 
The final design with the parallel polarization at 

the bottom of the PZT yielded a threshold voltage 

of 1.1𝑉 and an 𝐼𝐷,𝑚𝑎𝑥 of 6.9 × 10−5𝐴/µ𝑚 with a 

drain voltage of 0.25𝑉. The leakage current with 

this design is 1.9 × 10−10𝐴/µ𝑚, therefore 

negligible compared to the on-current. This can 

be seen in figure 5. This design compares 

favourably to a conventional HEMT with the 

same polarization between the AlGaN and the 

GaN. The conventional HEMT yields a lower 

𝐼𝐷,𝑚𝑎𝑥 of 5.6 × 10−5𝐴/µ𝑚 with a threshold 

voltage of −2.4𝑉, which can be seen in figure 6. 



When looking at the drain voltage with the same 

gate overdrive voltage however, the conventional 

HEMT and the final design yield the same drain 

current. Take a gate overdrive voltage of 2𝑉, here 

both the final design and the conventional HEMT 

have a drain current of 4.6 × 10−5𝐴/µ𝑚, which 

indicates that the design has a minimal if any 

impact on the drain current. This indicates that  

Note that the current collapses around the 1𝑉 

point, which is due to gate leakage (6). The 

parameters can be found in table 1. 

 

Figure 5: Drain current of the final design. 

 

Figure 6: Drain current of a conventional AlGaN/GaN HEMT. 

When looking at the band diagram, it is clearly 

visible that the conduction band is raised above 

the electron QFL (quasi Fermi level), meaning 

that the 2DEG is depleted. When the gate voltage 

is increased to past the threshold voltage, the 

conduction band dips below the electron QFL just 

below the AlGaN/GaN interface, therefore 

restoring the 2DEG under the gate. This can be 

seen in figure 7, while the state at 0𝑉 can be seen 

above in figure 3. The location of the 2DEG is just 

after the last black line in both figures. The 

electron and hole QFLs are (almost) equal, and 

therefore lay on top of each other. It is visible 

that the conduction band (red) dips below the 

electron QFL (green, but it is the same as the blue 

line) at the on state, but the conduction band is 

well above the Fermi level at the 0𝑉 state. 

 

Figure 7: Band diagram when conducting (7𝑉𝐺𝑎𝑡𝑒). 

4. Discussion 
The final design has a few caveats, this is because 

the polarization direction of the PZT layer seems 

to have a limited impact. The impact that it does 

have is a trade-off, a higher sheet charge, means 

a lower threshold voltage, but also a lower drain 

leakage current. Parallel polarization lowers the 

conduction band, therefore lowering the 

threshold voltage, which is the logical part of this 

trade-off. The band diagram of the antiparallel 

polarization can be seen in figure 4, which can be 

compared to the band diagram with parallel 

polarization in figure 3. To fully suppress the 

leakage current, a high enough parallel 

polarization is needed, this is because when this 

is high enough, the electron concentration in the 

AlGaN and GaN get forced out of the region 

under the gate stack significantly more than 

when there is a leakage current present, see 

figure 8. On the left side the electron 

concentration under the gate is significantly 

reduced, while this is not the case on the right. 

The resulting drain current with antiparallel 

polarization can be seen in figure 9. It is visible 

that the threshold voltage is significantly higher 

than with parallel polarization, and so is the 

leakage current. The device parameters of these 

two can be found in table 1. 

 

Figure 8: Electron concentration at 0𝑉𝐺𝑎𝑡𝑒 , with parallel 
(left) and antiparallel (right) polarization. 



 

Figure 9: Drain current with antiparallel polarization. 

The hypothesis why this is the case is that due to 

the antiparallel polarization, the hole 

concentration under the hafnium dioxide/AlGaN 

interface is increased. This increased hole 

concentration screens the 2DEG from the effects 

of the gate stack. The result is that the 2DEG is 

not fully depleted. Therefore, there is still a 

significant current flowing when the gate voltage 

is lower than the threshold voltage. Figure 10 

shows the difference in hole concentration in the 

top of AlGaN, figure 11 shows the electron 

concentration at the gate region, and figure 12 

shows the current flowlines, demonstrating that 

the current does indeed flow through the 2DEG 

and not deeper in the GaN. Note that the scale in 

this figure is on a relative scale, which means that 

the red area on the left does not have the same 

current flow as the same red on the right. One 

should only focus on the current flow in the GaN 

layer. 

 

Figure 10: Hole concentration in the top of the AlGaN, with 
parallel (left) and antiparallel (right) polarization. 

 

Figure 11: Electron concentration at the gate region, with 
parallel (left) and antiparallel (right) polarization. 

 

Figure 12: Current flowlines with a relative scale, with 
parallel (left) and antiparallel (right) polarization. 

When the leakage current is fully supressed, and 

there is the possibility to further increase the 

parallel polarization underneath the PZT, there is 

the option to increase the polarization at the 

AlGaN/GaN interface by increasing the aluminium 

content. This increase in polarization will increase 

the maximal drain current and makes it more 

difficult to deplete the 2DEG.  

Although there is still room to increase the drain 

current by increasing the polarization at the 

AlGaN/GaN interface, it was decided not to do 

this. This model has simplifications and 

approximations in it, and therefore will deviate 

from reality, where there will always be a leakage 

current, however small it may be. When there is 

more leakage current suppression due to a 

weaker 2DEG than may have worked without any 

leakage current in the model, the leakage current 

should be less in a real version as well. Another 

positive consequence of not going right to the 

edge of what works is leaving some wiggle room 

to adjust most if not all parameters to mitigate 

unforeseen flaws that may come up when less 

simplifications or for the reality.  

One of the significant simplifications is that the 

silicon oxide is an ideal isolator in the model, 

which is obviously not the case in real life, 

meaning that this has an impact on the 

functioning of the design. 

One peculiarity with the simulations is that the 

hafnium dioxide layer seems to act like an 

isolator instead of its intended use as a charge 

trapping layer. Despite this, the design still seems 

to work. If this is a quirk from Atlas, then there is 

a chance this design will not work as intended, if 

this behaviour carries over to the real world, then 

this design should still work. As mentioned 

above, changing only the name of a material 



could result in a different result because Atlas 

treats it differently. This was also tried with the 

hafnium oxide, by changing it to silicon, which 

Atlas treats as a semiconductor. This did not yield 

any different results, however. 

5. Conclusions and 

recommendations 
To conclude, the device simulation predict that 

the design offers normally off behaviour, with the 

caveat of the model simplifications and atlas 

peculiarities that may influence the outcome 

significantly. At 1.1𝑉, the threshold voltage 

shifted well into the positive range, resulting in a 

normally off AlGaN/GaN HEMT. The simulated 

leakage current is less than 0.001% of what the 

𝐼𝐷,𝑚𝑎𝑥 is, therefore more than enough of a 

difference for this to be used in high power 

applications. 

Although this design works, there is still plenty of 

room for improvements and optimizations. Some 

applications may prefer for example a higher 

threshold voltage, higher drain current, or an 

even lower drain current. Replacing the PZT 

and/or hafnium dioxide could be one way to 

reach the desired alteration. Another aspect 

where there is still room to further enhance the 

design is by optimizing the layer thicknesses, 

which is something that was done in the 

beginning, but that was before a normally off 

result was achieved, thus there is still room for 

optimization there as well. Reintroducing a 

partially recessed gate is also an option to 

increase the threshold voltage, although this will 

negatively affect the drain current (3). 

7. Appendix 

Larger figures 

Band diagram with parallel polarization (figure 3). 

 

 

Band diagram with anitparallel polarization (figure 4). 

  



 

Band diagram with parallel polarization at 7𝑉𝐺𝑎𝑡𝑒 (figure 7). 

 

 

Drain current conventional HEMT (figure 6). 

  



 

Drain current final design (figure 5). 

 

 

Drain current final design (figure 5), logarithmic y-axis. 

  



 

Drain current with antiparallel polarization (figure 9). 

 

 

Drain current with antiparallel polarization (figure 9), logarithmic y-axis. 

  



 

Electron concentration at 0𝑉𝐺𝑎𝑡𝑒 , with parallel (left) and antiparallel (right) polarization (figure 8). 

 

 

Hole concentration in the top of the AlGaN 0𝑉𝐺𝑎𝑡𝑒 , with parallel (left) and antiparallel (right) polarization 

(figure 10). 

  



 

Electron concentration at the gate region at 0𝑉𝐺𝑎𝑡𝑒 , with parallel (left) and antiparallel (right) polarization 

(figure 11). 

 

 

Current flowlines with a relative scale, with parallel (left) and antiparallel (right) polarization (figure 12). 



Input file 
go atlas 

 

# AlGaN/GaN HEMT, set layer thicknesses 

set tPZT=0.06 

set tALGAN=0.02 

set tHfO2=0.03 

 

# pz refers to the net polarization charge at each interface, setting sheet charge values 

set pz=0.8e13 

set pzPZT=8e13 

set pzHfO2=-2.7e13 

set meshratio=0.5 

mesh space.mult=1.0 

 

set vgstep=0.1 

 

# Define mesh in horizontal direction 

x.mesh loc=0.00    spac=0.02 

x.mesh loc=1.00    spac=0.02 

x.mesh loc=3.00    spac=0.02 

x.mesh loc=5.00    spac=0.02 

x.mesh loc=8.00            spac=0.02 

x.mesh loc=9.00            spac=0.02 

 

# Define mesh in vertical direction 

y.mesh loc=-1*$tPZT   spac=($tPZT-$tHfO2)/20 

y.mesh loc=-0.5*($tPZT-$tHfO2)-$tHfO2 spac=($tPZT-$tHfO2)/20 

y.mesh loc=-0.1*($tPZT-$tHfO2)-$tHfO2 spac=($tPZT-$tHfO2)/20 

 

y.mesh loc=-1*$tHfO2 spac=$tHfO2/40 

y.mesh loc=-0.9*$tHfO2 spac=$tHfO2/20 

y.mesh loc=-0.5*$tHfO2 spac=$tHfO2/10 

y.mesh loc=-0.1*$tHfO2 spac=$tHfO2/20 

 

y.mesh loc=0                                       spac=$tPZT/200 

y.mesh loc=0.1*$tALGAN                             spac=$tALGAN/40 

y.mesh loc=0.5*$tALGAN                             spac=$tALGAN/10 

y.mesh loc=0.9*$tALGAN                             spac=$tALGAN/40 

y.mesh loc=$tALGAN                                 spac=$tALGAN/60 

y.mesh loc=1.1*$tALGAN                             spac=$tALGAN/10 

y.mesh loc=3.00                                    spac=0.5 

 

# Define several regions, e.g., insulator or semiconductor layers 

region num=1 material=oxide      

region num=2 material=germanium y.min=-1*$tPZT         y.max=-1*$tHfO2              x.min=3 x.max=5 

region num=3 material=HfO2      y.min=-1*$tHfO2        y.max=0                        x.min=3 x.max=5 

region num=4 material=algan    y.min=0                y.max=$tALGAN x.comp=0.2    x.min=1 x.max=8 

region num=5 material=gan       y.min=$tALGAN          y.max=1.5+$tALGAN                     

 

# Define electrodes 

electr name=gate   y.min=-1*$tPZT y.max=-1*$tPZT                x.min=3 x.max=5  

electr name=source   y.min=-1*$tPZT y.max=0.02+$tALGAN            x.min=0 x.max=1  

electr name=drain   y.min=-1*$tPZT y.max=0.02+$tALGAN            x.min=8 x.max=9  

electr name=substrate bottom 

 

# Put a certain amount of interface charge to mimic the effect of polarization differences 

interface charge=$pzPZT s.s   y.min=-1*$tHfO2        y.max=-1*$tHfO2          x.min=3 x.max=5 

interface charge=$pzHfO2 s.s   y.min=0       y.max=0        x.min=3 x.max=5 

interface charge=$pz    s.s   y.min=$tALGAN  y.max=$tALGAN    x.min=1 x.max=8 

 

# Switch on models (which have to be completed for mobility) 

models print srh fermi 

 

# Adjust material parameters 

material material=gan       eg300=3.43   affinity=4.1  permittivity=9.5 mun=1800  mup=10 NC300=2.23e18 NV300=4.62e19  vsatn=1.9e7  

material material=algan     eg300=3.832  affinity=3.84 permittivity=9.4 mun=20    mup=10 NC300=2.95e18 NV300=1.084e20 vsatn=1.1e7 

material material=germanium eg300=3.4    affinity=3.5  permittivity=600 mun=25    mup=25 NC300=1.4e20  NV300=3.4e20 



material material=HfO2      eg300=5.8    affinity=2.0  permittivity=25  mun=150   mup=150  NC300=2.8e19 NV300=1.04e19 

 

 

# Impact ionization needed for breakdown 

impact selb material=gan 

 

# Adjust workfunction Schottky gate; source and drain regions are in principle ohmic 

contact name=gate workfunction=4.96 

contact name=source workfunction=4.2 

contact name=drain workfunction=4.2 

 

# Variables which should be plotted in case of TonyPlot 

output con.band val.band band.param flowlines charge traps e.mob h.mob e.velocity ex.vel ey.velocity h.vel hx.vel hy.vel qss polar.charge 

 

# Indicate what numerical methodology is used for the calculations: first adopt Gummel method, after 5 iterations the Newton method 

method gummel newton gum.init=5 carriers=2 autonr 

 

# Start calculations at thermal equilibrium 

solve init 

 

# Save the solution in a structure file for studying the device properties, e.g., plotting band diagrams or conduction current 

save outfile=PZT_HEMT.str 

 

########################################### 

# Calculation of I-V curves 

########################################### 

 

# Solve equations for one charge carrier 

method gummel newton gum.init=5 carriers=2 autonr 

 

# Solve the equations for a fixed drain voltage 

solve vdrain=0.25 

 

# Make a log file for storing the I-V data 

log outfile=Id_Vgs.log no.trap 

 

solve vgate=-3 vstep=$vgstep vfinal=-3 name=gate 

save outfile=Id_Vgs.str 

 

solve vgate=-3+$vgstep vstep=$vgstep vfinal=0 name=gate 

save outfile=Id_Vgs2.str 

 

solve vgate=$vgstep vstep=$vgstep vfinal=15 name=gate 

save outfile=Id_Vgs3.str 

 

# Stop storing the I-V data for the Id-Vds curve 

tonyplot Id_Vgs.log Id_Vgs.str Id_Vgs2.str Id_Vgs3.str 

quit 


